



Designation: E1926 – 08 (Reapproved 2021)

Standard Practice for Computing International Roughness Index of Roads from Longitudinal Profile Measurements¹

This standard is issued under the fixed designation E1926; the number immediately following the designation indicates the year of original adoption or, in the case of revision, the year of last revision. A number in parentheses indicates the year of last reapproval. A superscript epsilon (ϵ) indicates an editorial change since the last revision or reapproval.

1. Scope

1.1 This practice covers the mathematical processing of longitudinal profile measurements to produce a road roughness statistic called the International Roughness Index (IRI).

1.2 The intent is to provide a standard practice for computing and reporting an estimate of road roughness for highway pavements.

1.3 This practice is based on an algorithm developed in the International Road Roughness Experiment sponsored by a number of institutions including the World Bank and reported in two World Bank Technical Papers (**1**, **2**).² Additional technical information is provided in two Transportation Research Board (TRB) papers (**3**, **4**).

1.4 The values stated in SI units are to be regarded as the standard. The inch-pound units given in parentheses are for information only.

1.5 *This standard does not purport to address all of the safety concerns, if any, associated with its use. It is the responsibility of the user of this standard to establish appropriate safety, health, and environmental practices and determine the applicability of regulatory limitations prior to use.*

1.6 *This international standard was developed in accordance with internationally recognized principles on standardization established in the Decision on Principles for the Development of International Standards, Guides and Recommendations issued by the World Trade Organization Technical Barriers to Trade (TBT) Committee.*

2. Referenced Documents

2.1 ASTM Standards:³

¹ This practice is under the jurisdiction of ASTM Committee E17 on Vehicle - Pavement Systems and is the direct responsibility of Subcommittee E17.33 on Methodology for Analyzing Pavement Roughness.

Current edition approved Feb. 1, 2021. Published February 2021. Originally approved in 1998. Last previous edition approved in 2015 as E1926 – 08 (2015). DOI: 10.1520/E1926-08R21.

² The boldface numbers given in parentheses refer to a list of references at the end of the text.

³ For referenced ASTM standards, visit the ASTM website, www.astm.org, or contact ASTM Customer Service at service@astm.org. For *Annual Book of ASTM Standards* volume information, refer to the standard's Document Summary page on the ASTM website.

E177 Practice for Use of the Terms Precision and Bias in ASTM Test Methods

E867 Terminology Relating to Vehicle-Pavement Systems

E950/E950M Test Method for Measuring the Longitudinal Profile of Traveled Surfaces with an Accelerometer-Established Inertial Profiling Reference

E1082 Test Method for Measurement of Vehicular Response to Traveled Surface Roughness

E1170 Practices for Simulating Vehicular Response to Longitudinal Profiles of Traveled Surfaces

E1215 Specification for Trailers Used for Measuring Vehicular Response to Road Roughness

E1364 Test Method for Measuring Road Roughness by Static Level Method

E1656/E1656M Guide for Classification of Automated Pavement Condition Survey Equipment

E2133 Test Method for Using a Rolling Inclinometer to Measure Longitudinal and Transverse Profiles of a Traveled Surface

3. Terminology

3.1 Definitions:

3.1.1 Terminology used in this practice conforms to the definitions included in Terminology E867.

3.2 Definitions of Terms Specific to This Standard:

3.2.1 *International Roughness Index (IRI)*, n —an index computed from a longitudinal profile measurement using a quarter-car simulation (see Practice E1170) at a simulation speed of 80 km/h (50 mph).

3.2.1.1 *Discussion*—IRI is reported in either metres per kilometre (m/km) or inches per mile (in./mile). (Note—1 m/km = 63.36 in./mile.)

3.2.2 *longitudinal profile measurement*, n —a series of elevation values taken at a constant interval along a wheel track.

3.2.2.1 *Discussion*—Elevation measurements may be taken statically, as with rod and level (see Test Method E1364) or inclinometer (see Test Method E2133), or dynamically, as with an inertial profiler (see Test Method E950/E950M).

3.2.3 *Mean Roughness Index (MRI)*, n —the average of the IRI values for the right and left wheel tracks.

3.2.3.1 *Discussion*—Units are in metres per kilometre or inches per mile.

3.2.4 *traveled surface roughness*—the deviations of a surface from a true planar surface with characteristics dimensions that affect vehicle dynamics, ride quality, dynamic loads, and drainage, for example, longitudinal profile, transverse profile, and cross slope.

3.2.5 *true International Roughness Index, n* —the value of IRI that would be computed for a longitudinal profile measurement with the constant interval approaching zero.

3.2.6 *wave number, n* —the inverse of wavelength.

3.2.6.1 *Discussion*—Wave number, sometimes called spatial frequency, typically has units of cycle/m or cycle/ft.

3.2.7 *wheel track, n* —a line or path followed by the tire of a road vehicle on a traveled surface.

4. Summary of Practice

4.1 The practice presented here was developed specifically for estimating road roughness from longitudinal profile measurements.

4.2 Longitudinal profile measurements for one wheel track are transformed mathematically by a computer program and accumulated to obtain the IRI. The profile must be represented as a series of elevation values taken at constant intervals along the wheel track.

4.3 The IRI scale starts at zero for a road with no roughness and covers positive numbers that increase in proportion to roughness. Fig. 1 associated typical IRI values with verbal descriptors from World Bank Technical Paper No. 46 (2) for roads with bituminous pavement, and Fig. 2 shows similar associations for roads with earth or gravel surfaces.

5. Significance and Use

5.1 This practice provides a means for obtaining a quantitative estimate of a pavement property defined as roughness using longitudinal profile measuring equipment.

5.1.1 The IRI is portable in that it can be obtained from longitudinal profiles obtained with a variety of instruments.

5.1.2 The IRI is stable with time because true IRI is based on the concept of a true longitudinal profile, rather than the physical properties of a particular type of instrument.

5.2 Roughness information is a useful input to the pavement management systems (PMS) maintained by transportation agencies.

5.2.1 The IRI for the right wheel track is the measurement of road surface roughness specified by the Federal Highway Administration (FHWA) as the input to their Highway Performance Monitoring System (HPMS).

5.2.2 When profiles are measured simultaneously for both traveled wheel tracks, then the MRI is considered to be a better measure of road surface roughness than the IRI for either wheel track.

NOTE 1—The MRI scale is identical to the IRI scale.

5.3 IRI can be interpreted as the output of an idealized response-type measuring system (see Test Method E1082 and Specification E1215), where the physical vehicle and instrumentation are replaced with a mathematical model. The units

of slope correspond to accumulated suspension motions (for example, metres), divided by the distance traveled (for example, kilometres).

5.4 IRI is a useful calibration reference for response-type systems that estimate roughness by measuring vehicular response (see Test Method E1082 and Specification E1215).

5.5 IRI can also be interpreted as average absolute slope of the profile, filtered mathematically to modify the amplitudes associated with different wavelengths (3).

6. Longitudinal Profile Measurement

6.1 The longitudinal profile measurements can be obtained from equipment that operates in a range of speeds from static to highway traffic speeds.

6.2 The elevation profile measuring equipment used to collect the longitudinal profile data used in this practice must have sufficient accuracy to measure the longitudinal profile attributes that are essential to the computation of the IRI.

7. Computation of International Roughness Index (IRI)

7.1 This practice consists of the computation of IRI from an algorithm developed in the International Road Roughness Experiment and described in the World Bank Technical Papers 45 and 46 (1, 2). Additional technical information provided in two TRB papers (3, 4).

7.2 A Fortran version of this algorithm has been implemented as described in Ref (3).

7.2.1 This practice presents a sample computer program “IRISMP” for the computation of the IRI from the recorded longitudinal profile measurement.

7.2.1.1 The computer program IRISMP is a general computer program which accepts the elevation profile data set as input and then calculates the IRI values for that profile data set.

7.2.1.2 A listing of the IRISMP computer program for the computation of IRI is included in this practice as Appendix X2.

7.2.1.3 A provision has been made in the computer program listing (Appendix X2) for the computation of IRI from recorded longitudinal profile measurements in either SI or inch-pound units.

7.2.2 The input to the sample IRI computer program is an ASCII profile data set stored in a 1X,F8.3,1X,F8.3 Fortran format. In this format, the profile data appear as a multi-row, two column array with the left wheel path profile data points in Column 1 and the right wheel path points in Column 2. The profile data point interval is discretionary. However the quality of the IRI values computed by this algorithm is a function of the data point interval.

7.2.2.1 If the input to the IRI computer program is in SI units, the elevation profile data points are scaled in millimetres with the least significant digit being equal to 0.001 mm.

7.2.2.2 If the input to the IRI computer program is in inch-pound units, the elevation profile data points are scaled in inches with the least significant digit being equal to 0.001 in.

7.3 The distance interval over which the IRI is computed is discretionary, but shall be reported along with the IRI results.




FIG. 1 Road Roughness Estimation Scale for Paved Roads with Asphaltic Concrete or Surface Treatment (Chipseal)

7.4 Validation of the IRI program is required when it is installed. Provision for the IRI program installation validation has been provided in this practice.

7.4.1 The sample profile data set TRIPULSE.ASC has been provided in SI units in [Appendix X2](#) for validation of the computer program installation.

7.4.2 Using the sample profile data set TRIPULSE.ASC as input to the IRI computer program, an IRI value of 4.36 mm/m was computed for a profile data point interval of 0.15 m (0.5 ft) and a distance interval equal to 15 m of the profile data set in [Appendix X2](#).

8. Report

8.1 Include the following information in the report for this practice:

8.1.1 *Profile Measuring Device*—The class of the profile measuring device used to make the profile measurement as defined in Test Method [E950/E950M](#) and Test Method [E1364](#) shall be included in the report.

8.1.2 *Longitudinal Profile Measurements*—Report data from the profile measuring process shall include the date and time of day of the measurement, the location of the measurement, the

ROUGHNESS
(m/km IRI)




FIG. 2 Road Roughness Estimation Scale for Unpaved Roads with Gravel or Earth Surfaces

lane measured, the direction of the measurement, length of measurement, and the descriptions of the beginning and ending points of the measurement. The recorded wheel track (left, right, or both) must also be included.

8.1.3 *IRI Resolution*—The number of digits after the decimal point depends on the choice of units. If the units are m/km, then results should be reported with two digits after the decimal

point. If the units are in./mile, then the IRI results should be reported to a resolution of 0.1 in./mile.

9. Precision and Bias

9.1 The precision and bias of the computed IRI is limited by the procedures used in making the longitudinal profile measurement. Guidelines for measuring longitudinal profile are provided in Test Method **E950/E950M** and Test Method **E1364**.

9.2 For the effects of the precision and bias of the measured profile on the computed IRI, see precision and bias in **Appendix X1**.

10. Keywords

10.1 highway performance monitoring system; HPMS; international roughness index; International Roughness Index; longitudinal profile; pavement management systems; pavement roughness; PMS

APPENDIXES

(Nonmandatory Information)

X1. PRECISION AND BIAS

X1.1 *Precision:*

X1.1.1 The precision of the computed IRI is limited by the procedures used in making the longitudinal profile measurement. Guidelines for measuring longitudinal profile are provided in Test Method **E950/E950M** and Test Method **E1364**.

X1.1.2 IRI precision depends on the interval between adjacent profile elevation measures (see Test Method **E950/E950M** and Test Method **E1364**). Reducing the interval typically improves the precision. An interval of 0.3 m (12 in.) or smaller is recommended. For some surface types, a shorter interval will improve precision. More information about the sensitivity of IRI to the profile data interval is provided in Ref (3).

X1.1.3 IRI precision is roughly equivalent to the precision of the slope obtained from the longitudinal profile measurements, for distances ranging from approximately 1.5 m (5 ft) to about 25 m (80 ft). For example, a relative error on profile elevation of 1.0 mm over a distance of 10 m corresponds to a slope error of 0.1 mm/m, or 0.1 m/km (6.3 in./mi).

X1.1.4 IRI precision is limited by the degree to which a wheel track on the road can be profiled. Errors in locating the wheel track longitudinally and laterally can influence the IRI values, because the IRI will be computed for the profile of the wheel track as measured, rather than the wheel track as intended. These effects are reduced by using longer profiles.

X1.1.5 Computational errors due to round-off are typically about two orders of magnitude smaller than those due to limitations in the profile measuring process, and can be safely ignored.

X1.2 *Bias:*

X1.2.1 The bias of the computed IRI is typically limited by the procedures used in making the longitudinal profile measurement. Guidelines for measuring longitudinal profile are provided in Test Method **E950/E950M** and Test Method **E1364**.

X1.2.2 IRI bias depends on the interval between adjacent profile elevation measures. An interval of 0.3 m (12 in.) or smaller is recommended. Shorter intervals improve precision but have little effect on bias. More information about the sensitivity of IRI to the profile data interval is provided in Ref (3).

X1.2.3 Many forms of measurement error cause an upward bias in IRI. (The reason is that variations in profile elevation due to measurement error are usually not correlated with the profile changes.) Some common sources of positive IRI bias are: height-sensor round-off, mechanical vibrations in the instrument that are not corrected and electronic noise. Bias is reduced by using profiler instruments that minimize these errors.

X1.2.4 Inertial profiler systems (see Test Method **E950/E950M**) include one or more filters that attenuate long wavelengths (low wave numbers). If the cut-off wavelength is too short, then the IRI computed from the profile will have a negative bias. A cut off wavelength of 91.4 m/cycle (300 ft/cycle) is considered sufficiently long.

NOTE X1.1—Profiles obtained with static methods are generally not filtered, and therefore this source of bias is not relevant for them.

X1.2.5 The measures from some inertial profilers are processed during measurement to attenuate short wavelengths and prevent aliasing. The effect is to smooth the profile measurement. If a smoothing filter is used and it affects wavelengths longer than 1 m (3.3 ft), then the computed IRI will have a negative bias.

NOTE X1.2—If the profiler includes a smoothing filter that affects wavelengths shorter than 1 m (3.3 ft) and longer than 250 mm (10 in.), no more smoothing is required during the computation of IRI.

X2. INTERNATIONAL ROUGHNESS INDEX COMPUTER PROGRAM

X2.1 Included in this appendix is the coding in Fortran language for a computer subroutine, SUBROUTINE IRI, (see Fig. X2.1), which calculates the International Roughness Index

as prescribed by this practice. A sample main program is also included, which when executed, prompts the user for the name of a data file containing the profile data to be processed and the

```

C=====
C  Sample IRI Fortran Computer Program
C
C  Sample program to read a data file containing two tracks of road
C  profile elevation data into a "DATA" array, call SUBROUTINE IRI
C  and print a final report of International Roughness Index.
C  If the input profile data are in English units, the elevation values
C  are converted from inch to mm units and the sampling interval, from
C  feet to meters; the computed IRI values are returned as m/km and
C  converted to in/mi.
C  (SUBROUTINE IRI is called to perform the IRI computation as
C  prescribed by this practice.)
C
      PROGRAM IRISMP

C  DELT          --> DX
C  PROFL(1058)   --> left track profile
C  PROFR(1058)   --> right track profile
C  AVEIRIL       --> IRI, left track
C  AVEIRIR       --> IRI, right track
C  AVEIRI        --> == (AVEIRIL+AVEIRIR)/2.
C  UNITSC        --> see UNITSC (SUBR IRI)

      REAL      DELT, SECLN
      REAL      BASE, UNITSC, PROFL(1058), PROFR(1058)
      REAL      AVIRIL, AVIRIR, AVEIRI

      BYTE      ANSWER
      CHARACTER  KNAME*12

      INTEGER    NPTS, NREC, I
      NREC = 0

      WRITE(*,1000)
1000 FORMAT(/'Enter data file name (in single quotes)'/
1      '      ("TRIPULSE.ASC" in example):  '$)

      READ(*,*) KNAME(1:12)

      WRITE(*,1010)
1010 FORMAT(/'Enter the number of samples in the profile.'/
1      '      (101 in example) :  '$)

```

FIG. X2.1 Sample Fortran Program Using Subroutine IRI to Compute International Roughness Index

```

      READ(*,1020) NPTS
1020 FORMAT(I8)

      WRITE(*,1030)
1030 FORMAT(/'Enter the sampling interval, meters'/
1      '      (.15 m in example) :  '$)

      READ(*,1040) DELT
1040 FORMAT(F10.0)

      SECLN = REAL(NPTS-1)*DELT

      WRITE(*,1050)
1050 FORMAT(/'Is the input profile pre-smoothed (Y or N)? '$)
      READ(*,1060) ANSWER
1060 FORMAT(A1)
      BASE = .250
      IF(ANSWER.NE.'N'.AND.ANSWER.NE.'n') BASE = 0.

c-Open input file and read profile elevations into 'PROF' arrays:

      OPEN(UNIT=2,FILE=KNAME(1:12),FORM='FORMATTED')

      UNITSC = 1.

      DO 20 I = 1,NPTS
20      READ(2,1070) PROFL(I), PROFR(I)
1070 FORMAT(2(1X,F8.3))

c-Call subroutine to calculate International Roughness Index:
      NSAMP = NPTS
      CALL IRI(PROFL,NSAMP,DELT,BASE,UNITSC,AVIRIL)
      NSAMP = NPTS
      CALL IRI(PROFR,NSAMP,DELT,BASE,UNITSC,AVIRIR)

      AVEIRI=(AVIRIL + AVIRIR)/2.

      WRITE(*,2020)
1 AVIRIL, AVIRIR, AVEIRI, SECLN

2020 FORMAT(////////
1      6X'IRI, left          = ',F10.2,' m/km'//
2      6X'IRI right         = ',F10.2,' m/km'////
3      6X'International Roughness Index = ',F10.2,' m/km'//
4      6X' Distance          = ',F6.1,' meters'//)
      END

```

FIG. X2.1 Sample Fortran Program Using Subroutine IRI to Compute International Roughness Index *(continued)*

parameters needed by the subroutine to compute the IRI. The subroutine is called and returns the computed IRI values to the main program which then displays them.

X2.2 The sample program can process data files containing two profile tracks in either SI or inch-pound units. For SI data, the program assumes the input amplitudes are stored in millimetre units; if inch-pound, inches. For the sample

program, the maximum length road section that can be processed is limited to 1058 sample pairs.

X2.3 The sample data file shown in [Fig. X2.2](#) and [Fig. X2.3](#) is in SI units (mm) and contains 101 profile data point pairs. The tracks are identical. The recording interval for the data is 0.15 m.


```

C=====
      SUBROUTINE IRI(PROF, NSAMP, DX, BASE, UNITSC, AVEIRI)
C=====
C Filter a longitudinal road profile and calculate IRI.
C
C <--> PROF    REAL      On input, an array of profile height values.
C                               On output, an array of filtered profile values.
C <--> NSAMP    INTEGER    Number of data value in array PROF. Filtered
C                               profile always has fewer points than original.
C --> DX       REAL      Distance step between profile points (m).
C --> BASE     REAL      Distance covered by moving average (m).
C                               Use .250 for unfiltered profile input, and 0.0
C                               for pre-smoothed profiles (e.g. K.J. Law data).
C --> UNITSC   REAL      Product of two scale factors: (1) meters per unit
C                               of profile height, and (2) IRI units of slope.
C                               Ex: height is inches, slope will be in/mi.
C                               UNITSC = (.0254 m/in)*(63360 in/mi) = 1069.34
C <-- AVEIRI   REAL      The average IRI for the entire profile.

      INTEGER    I, I11, IBASE, NSAMP
      REAL       AMAT, AVEIRI, BASE, BMAT, CMAT, DX
      REAL       UNITSC, XIN, PROF, SFPI, ST, PR
      DIMENSION AMAT(4, 4), BMAT(4), CMAT(4), PR(4),
&              ST(4,4), XIN(4), PROF(NSAMP)

C Set parameters and arrays.
      CALL SETABC(653.0, 63.3, 6.0, 0.15, AMAT, BMAT, CMAT)
      CALL SETSTM(DX/(80./3.6), AMAT, BMAT, ST, PR)
      IBASE = MAX(INT(BASE/DX + 0.5), 1)
      SFPI = UNITSC/(DX*IBASE)

C Initialize simulation variables based on profile start.
      I11 = MIN(INT(11./DX + 0.5) + 1, NSAMP)
      XIN(1) = UNITSC*(PROF(I11) - PROF(1))/(DX*I11)
      XIN(2) = 0.0
      XIN(3) = XIN(1)
      XIN(4) = 0.0

C Convert to averaged slope profile, with IRI units.
      NSAMP = NSAMP - IBASE
      DO 10 I = 1, NSAMP
10    PROF(I) = SFPI*(PROF(I + IBASE) - PROF(I))

```

FIG. X2.1 Sample Fortran Program Using Subroutine IRI to Compute International Roughness Index *(continued)*


```

C  Filter profile.
      CALL STFILT(PROF, NSAMP, ST, PR, CMAT, XIN)

C  Compute IRI from filtered profile.
      AVEIRI = 0.0
      DO 20 I = 1, NSAMP
20    AVEIRI = AVEIRI + ABS(PROF(I))
      AVEIRI = AVEIRI/NSAMP
      RETURN
      END

C=====
      SUBROUTINE SETABC(K1, K2, C, MU, AMAT, BMAT, CMAT)
C=====
C  Set the A, B and C matrices for the 1/4 car model.
C
C -->  K1      REAL    Kt/Ms = normalized tire spring rate, (1/s/s)
C -->  K2      REAL    Ks/Ms = normalized suspension spring rate (1/s/s)
C -->  C       REAL    C/Ms   = normalized suspension damper rate (1/s)
C -->  MU      REAL    Mu/Ms = normalized unsprung mass (-)
C <--  AMAT    REAL    The 4x4 A matrix.
C <--  BMAT    REAL    The 4x1 B matrix.
C <--  CMAT    REAL    The 4x1 C matrix.

      INTEGER      I, J
      REAL          AMAT, BMAT, CMAT, K1, K2, C, MU
      DIMENSION     AMAT(4, 4), BMAT(4), CMAT(4)

C  Set default for all matrix elements to zero.
      DO 10 J = 1, 4
        BMAT(J) = 0
        CMAT(J) = 0
        DO 10 I = 1, 4
10      AMAT(I, J) = 0

C  Put 1/4 car model parameters into the A Matrix.
      AMAT(1, 2) = 1.
      AMAT(3, 4) = 1.
      AMAT(2, 1) = -K2
      AMAT(2, 2) = -C
      AMAT(2, 3) = K2
      AMAT(2, 4) = C
      AMAT(4, 1) = K2/MU
      AMAT(4, 2) = C/MU
      AMAT(4, 3) = -(K1 + K2)/MU
      AMAT(4, 4) = -C/MU

```

FIG. X2.1 Sample Fortran Program Using Subroutine IRI to Compute International Roughness Index (continued)


```

C Set the B matrix for road input through tire spring.
  BMAT(4) = K1/MU

C Set the C matrix to use suspension motion as output.
  CMAT(1) = -1
  CMAT(3) = 1

  RETURN
  END

C=====
  SUBROUTINE SETSTM(DT, A, B, ST, PR)
C=====
C
C Compute ST and PR arrays. This requires INVERT for matrix inversion.
C
C --> DT      REAL    Time step (sec)
C --> A       REAL    The 4x4 A matrix.
C --> B       REAL    The 4x1 B matrix.
C <-- ST      REAL    4x4 state transition matrix.
C <-- PR      REAL    4x1 partial response vector.

  INTEGER      I, ITER, J, K
  LOGICAL      MORE
  REAL         A, A1, A2, B, DT, PR, ST, TEMP
  DIMENSION A(4, 4), A1(4, 4), A2(4, 4), B(4)
  DIMENSION PR(4), ST(4, 4), TEMP(4, 4)

  DO 20 J = 1, 4
    DO 10 I = 1, 4
      A1(I, J) = 0
10    ST(I, J) = 0
      A1(J, J) = 1.
20    ST(J, J) = 1.

C Calculate the state transition matrix ST = exp(dt*A) with a Taylor
C series. A1 is the previous term in the series, A2 is the next one.
  ITER = 0
30 ITER = ITER + 1
  MORE = .FALSE.
  DO 40 J = 1, 4
    DO 40 I = 1, 4
      A2(I, J) = 0
      DO 40 K = 1, 4
40    A2(I, J) = A2(I, J) + A1(I, K)*A(K, J)

```

FIG. X2.1 Sample Fortran Program Using Subroutine IRI to Compute International Roughness Index (continued)


```

      DO 50 J = 1, 4
        DO 50 I = 1, 4
          A1(I, J) = A2(I, J)*DT/ITER
          IF (ST(I, J) + A1(I, J) .NE. ST(I, J)) MORE = .TRUE.
50      ST(I, J) = ST(I, J) + A1(I, J)
      IF (MORE) GO TO 30

C  Calculate particular response matrix: PR = A**-1*(ST-I)*B
      CALL INVERT(A, 4)
      DO 60 I = 1, 4
        PR(I) = 0.0
        DO 60 K = 1, 4
60      PR(I) = PR(I) - A(I, K)*B(K)
      DO 90 J = 1, 4
        DO 70 I = 1, 4
          TEMP(J, I) = 0.0
          DO 70 K = 1, 4
70      TEMP(J, I) = TEMP(J, I) + A(J, K)*ST(K, I)
        DO 80 K = 1, 4
80      PR(J) = PR(J) + TEMP(J, K)*B(K)
90      CONTINUE
      RETURN
      END

C=====
      SUBROUTINE STFILT(PROF, NSAMP, ST, PR, C, XIN)
C=====
C  Filter profile using matrices ST, PR, and
C
C  <--> PROF   REAL      Input profile. Replaced by the output.
C  --> NSAMP  INTEGER    Number of data values in array PROF.
C  --> ST     REAL      4x4 state transition matrix.
C  --> PR     REAL      4x1 partial response vector.
C  --> C      REAL      4x1 output definition vector.
C  --> XIN    REAL      Initial values of filter variables.

      INTEGER    I, J, K, NSAMP
      REAL       C, PR, PROF, ST, X, XIN, XN
      DIMENSION C(4), PR(4), PROF(NSAMP), ST(4, 4), X(4), XIN(4), XN(4)

C  Initialize simulation variables.
      DO 10 I = 1, 4
10      X(I) = XIN(I)

```

FIG. X2.1 Sample Fortran Program Using Subroutine IRI to Compute International Roughness Index (continued)


```

C  Filter profile using the state transition algorithm.
      DO 40 I = 1, NSAMP
        DO 20 J = 1, 4
          XN(J) = PR(J)*PROF(I)
          DO 20 K = 1, 4
            20      XN(J) = XN(J) + X(K)*ST(J, K)
          DO 30 J = 1, 4
            30      X(J) = XN(J)
          PROF(I) = X(1)*C(1) + X(2)*C(2) + X(3)*C(3) + X(4)*C(4)
        40      CONTINUE
      RETURN
      END

C=====
      SUBROUTINE INVERT(Y1, N)
C=====
C  This routine will store the inverse of NxN matrix Y1 in matrix YINV.
C  It was copied from "Numerical Recipes."
C
C  Y1    --> Real      The matrix to be inverted.
C  YINV  --> Real      The inverse of matrix Y1.
C
      INTEGER      N, INDX, I, J
      REAL*4       Y1, YINV, D, A
      DIMENSION    Y1(N, N), YINV(4, 4), INDX(4), A(4, 4)

      DO 8 I = 1, N
        DO 9 J = 1, N
          9      A(I, J) = Y1(I, J)
        8      CONTINUE
      DO 10 I = 1, N
        DO 20 J = 1, N
          20      YINV(I, J) = 0.0
          YINV(I, I) = 1.0
        10      CONTINUE
      CALL LUDCMP(A, INDX, D)
      DO 30 J = 1, N
        30      CALL LUBKSB(A, INDX, YINV(1, J))
      DO 40 I = 1, N
        DO 50 J = 1, N
          50      Y1(I, J) = YINV(I, J)
        40      CONTINUE
      RETURN
      END

```

FIG. X2.1 Sample Fortran Program Using Subroutine IRI to Compute International Roughness Index *(continued)*


```

C=====
      SUBROUTINE LUDCMP(A, INDX, D)
C=====
C  This routine was copied from "Numerical Recipes" for matrix
C  inversion.
C
      INTEGER      N, INDX, NMAX, I, J, IMAX, K
      REAL*4       A, TINY, VV, D, AAMAX, SUM, DUM
      PARAMETER    (NMAX = 100, TINY = 1.0E-20, N = 4)
      DIMENSION    A(N, N), INDX(N), VV(NMAX)

      D = 1.0
      DO 10 I = 1, N
        AAMAX = 0.0
        DO 20 J = 1, N
          20  IF(ABS(A(I,J)).GT.AAMAX) AAMAX=ABS(A(I,J))
          IF(AAMAX.EQ.0.0) PAUSE 'Singular matrix'
          VV(I) = 1.0/AAMAX
        10  CONTINUE
        DO 30 J = 1, N
          DO 40 I = 1, J-1
            SUM = A(I, J)
            DO 50 K = 1, I-1
              50  SUM = SUM - A(I, K)*A(K, J)
            A(I, J) = SUM
          40  CONTINUE
          AAMAX = 0.0
          DO 60 I = J, N
            SUM = A(I, J)
            DO 70 K = 1, J-1
              70  SUM = SUM - A(I, K)*A(K, J)
            A(I, J) = SUM
            DUM = VV(I)*ABS(SUM)
            IF(DUM.GE.AAMAX) THEN
              IMAX = I
              AAMAX = DUM
            ENDIF
          60  CONTINUE
          IF(J.NE.IMAX) THEN
            DO 80 K = 1, N
              DUM = A(IMAX, K)
              A(IMAX, K) = A(J, K)
              A(J, K) = DUM
            80  CONTINUE
            D = -D
          
```

FIG. X2.1 Sample Fortran Program Using Subroutine IRI to Compute International Roughness Index (continued)


```

        VV(IMAX) = VV(J)
    ENDIF
    INDX(J) = IMAX
    IF(A(J, J).EQ.0.0) A(J, J) = TINY
    IF(J.NE.N) THEN
        DUM = 1.0/A(J, J)
        DO 90 I = J+1, N
90      A(I, J) = A(I, J)*DUM
        ENDIF
30    CONTINUE
    RETURN
    END
C=====
      SUBROUTINE LUBKSB(A, INDX, B)
C=====
C  This routine was copied from "Numerical Recipes" for matrix
C  inversion.

      INTEGER      N, INDX, I, II, LL, J
      REAL*4       A, B, SUM
      PARAMETER    (N = 4)
      DIMENSION    A(N, N), INDX(N), B(N)

      II = 0
      DO 10 I = 1, N
          LL = INDX(I)
          SUM = B(LL)
          B(LL) = B(I)
          IF(II.NE.0) THEN
              DO 20 J = II, I-1
20          SUM = SUM - A(I, J)*B(J)
              ELSEIF(SUM.NE.0) THEN
                  II = I
              ENDIF
          B(I) = SUM
10      CONTINUE
      DO 30 I = N, 1, -1
          SUM = B(I)
          IF(I.LT.N) THEN
              DO 40 J = I+1, N
40          SUM = SUM - A(I, J)*B(J)
              ENDIF
          B(I) = SUM/A(I, I)
30      CONTINUE
    RETURN
    END

```

FIG. X2.1 Sample Fortran Program Using Subroutine IRI to Compute International Roughness Index (continued)

0.000	0.000
0.000	0.000
0.000	0.000
0.000	0.000
0.000	0.000
2.500	2.500
5.000	5.000
7.500	7.500
10.000	10.000
12.500	12.500
15.000	15.000
17.500	17.500
20.000	20.000
17.500	17.500
15.000	15.000
12.500	12.500
10.000	10.000
7.500	7.500
5.000	5.000
2.500	2.500
0.000	0.000
0.000	0.000
0.000	0.000
0.000	0.000
0.000	0.000
0.000	0.000
0.000	0.000
0.000	0.000
0.000	0.000
0.000	0.000
0.000	0.000
0.000	0.000
...	(pad with zeros to make a total of 101 numerical data)
0.000	0.000
0.000	0.000
0.000	0.000
0.000	0.000
0.000	0.000
0.000	0.000

NOTE 1—Elevations are metric units (mm). The profile consists of identical right and left wheel tracks, each consisting of zero elevations everywhere except the triangular ‘pulse’ from 0.6 to 3.0 m peaking at 20.0 mm. The interval between elevations is 0.15 m and the total length is 15 m. This data set may be used as a test of the user’s implementation of IRI standard computation.

FIG. X2.2 Sample Load Profile Input Data Set, TRIPULSE.ASC


```
1Enter data file name (in single quotes)
  ("TRIPULSE.ASC" in example):  "TRIPULSE.ASC"
```

```
Enter the number of samples in the profile.
(101 in example) :  101
```

```
Enter the sampling interval, meters
(.15 m in example) :  .15
```

```
Is the input profile pre-smoothed (Y or N)? N
```

```
IRI, left          =          4.36 m/km
```

```
IRI right         =          4.36 m/km
```

```
International Roughness Index =          4.36 m/km
```

```
Distance          =   15.0 meters
```

FIG. X2.3 Input/Output for RNSMP sample program using data input file 'TRIPULSE.ASC'

REFERENCES

- | | |
|--|---|
| <p>(1) Sayers, M. W., Gillespie, T. D., Queiroz, C. A. V., "The International Road Roughness Experiment," <i>World Bank Technical Paper</i>, Number 45, 1986.</p> <p>(2) Sayers, M. W., Gillespie, T. D., Paterson, W. D. O., "Guidelines for Conducting and Calibrating Road Roughness Measurements," <i>World Bank Technical Paper</i>, Number 46, 1986.</p> | <p>(3) Sayers, M. W., "On the Calculation of IRI from Longitudinal Profile," <i>Transportation Research Record</i>, Vol 1501, Transportation Research Board, 1995, pp. 1–12.</p> <p>(4) Sayers, M. W., "Two Quarter-Car Models for Defining Road Roughness: IRI and HRI," <i>Transportation Research Record</i>, Vol 1215, 1989, pp. 165–172.</p> |
|--|---|

ASTM International takes no position respecting the validity of any patent rights asserted in connection with any item mentioned in this standard. Users of this standard are expressly advised that determination of the validity of any such patent rights, and the risk of infringement of such rights, are entirely their own responsibility.

This standard is subject to revision at any time by the responsible technical committee and must be reviewed every five years and if not revised, either reapproved or withdrawn. Your comments are invited either for revision of this standard or for additional standards and should be addressed to ASTM International Headquarters. Your comments will receive careful consideration at a meeting of the responsible technical committee, which you may attend. If you feel that your comments have not received a fair hearing you should make your views known to the ASTM Committee on Standards, at the address shown below.

This standard is copyrighted by ASTM International, 100 Barr Harbor Drive, PO Box C700, West Conshohocken, PA 19428-2959, United States. Individual reprints (single or multiple copies) of this standard may be obtained by contacting ASTM at the above address or at 610-832-9585 (phone), 610-832-9555 (fax), or service@astm.org (e-mail); or through the ASTM website (www.astm.org). Permission rights to photocopy the standard may also be secured from the Copyright Clearance Center, 222 Rosewood Drive, Danvers, MA 01923, Tel: (978) 646-2600; <http://www.copyright.com/>